

Le type « plage de cellules » spécifique à Excel

LE TYPE RANGE

Le type RANGE désigne une **plage de cellules**, c'est un type spécifique à Excel.

Exemple

Coin en haut et à gauche de la plage de cellules passée en paramètre de la fonction = coordonnée (1, 1) c.-à-d. ligne n°1 et colonne n°1, quelle que soit la position absolue de la plage dans la feuille de calcul (ici le coin nord-ouest est en B3)

The screenshot shows an Excel spreadsheet with columns A through F and rows 1 through 7. A range of cells from B3 to D4 is selected, highlighted in blue. The formula bar at the top displays the formula `=MaSommeRange(B3:D4)`. A red arrow points from the top-left corner of the selected range (cell B3) to the text box above. A purple arrow points from the bottom-right corner of the selected range (cell D4) to the text box below.

	A	B	C	D	E	F
1						
2						
3		12	10	7		
4		6	8	2		
5						
6						
7						

Un bloc de cellules (B3:D4) est passé en paramètre de la fonction. Ce bloc est forcément rectangulaire, avec, ici : 2 lignes et 3 colonnes.



La fonction `MaSommeRange()` est censée faire la même chose que la fonction standard `SOMME()` d'Excel.

Entrée : plage (range)
Sortie : S (réel)
Calcul : Somme des valeurs

Lignes et colonnes commencent à l'indice 1, quelle que soit la position de la plage dans la feuille.



```
'Travail sur le type Range
Public Function MaSommeRange(plage As Range) As Double
'variables intermédiaires
Dim s As Double, i As Long, j As Long
'initialisation de la somme
s = 0
'parcours de la plage de cellules
For i = 1 To plage.Rows.Count Step 1 'lignes
    For j = 1 To plage.Columns.Count Step 1 'colonnes
        'lecture des valeurs et somme
        s = s + plage.Cells(i, j).Value
    Next j
Next i
'renvoyer le résultat
MaSommeRange = s
End Function
```

Nombre de lignes de la plage de cellules.

Nombre de colonnes.

Accès à la valeur (Value) de la cellule : ligne n°i, colonne n°j

La boucle For Each est adaptée au parcours des collections.
Or une plage de cellules est une collection de cellules.

'Travail sur le type Range avec un For Each

```
Public Function MaSommeRangeEach(plage As Range) As Double
```

'variables intermédiaires

```
Dim s As Double, cellule As Range
```

'initialisation de la somme

```
s = 0
```

'parcours de la plage de cellules

```
For Each cellule In plage
```

```
    s = s + cellule.Value
```

```
Next cellule
```

'renvoyer le résultat

```
MaSommeRangeEach = s
```

```
End Function
```

Une cellule est une plage de cellules avec une seule cellule.

plage fait figure de collection à traiter.
Qu'importe le sens du parcours ici (ligne par ligne, ou colonne par colonne).

C'est bien la valeur contenue dans la cellule qui est exploitée pour la somme.